


APPLICATION NOTE 377-3 FREQUENCY PROFILING WITHOUT A PULSE GENERATOR

Using the HP 5361A 20 GHz
Pulse/CW Microwave Counter's
Built-In Profiling



High Resolution Pulse Profiling . . . without a Pulse Generator

■ A revolutionary new profiling technique invented by Hewlett-Packard now makes it possible to profile frequency pulses without the need for an external pulse generator. Called Built-In Profiling, this technique is featured in the HP 5361A 20 GHz Pulse/CW Microwave Counter and enables profiling of pulsed signals using gate widths as narrow as 10.5 nsec.

Figure 1 shows the results of profiling a 4 MHz frequency step on a 5.3 GHz carrier using only the HP 5361A and a technical computer.




Figure 1. Built-In Profiling of a 4 MHz step on a 5.3 GHz carrier.

Built-In Profiling

■ This new pulse profiling method requires only an HP 5361A counter and a computer with HP-IB capability. While traditional profiling methods require an external delaying pulse generator, Built-In Profiling makes use of the counter's internal gating functions in a unique way.

Built-In Profiling offers several distinct advantages over other, more conventional profiling methods:

- No external delay generator is needed
- Gate widths are as narrow as 10.5 nsec
- Higher frequency accuracy is possible
- Jitter in gate position is eliminated
- Profiling is quick and easy

Event Gating

The HP 5361A provides four main methods of gating a measurement: external gating, automatic internal gating, time gating, and event gating. (See the HP 5361A Data Sheet (5952-7986) or Application Note 377-1 (5952-7987) for more details on the first three methods.) It is the powerful capabilities of event gating that enable Built-In Pulse Profiling.

Single Event Measurement

When used with a controller and the BASIC program described on pages 4-8, the HP 5361A first automatically acquires the pulsed signal, determines the pulse width and average pulse frequency, and begins the gating, or profiling, process. During this process, the frequency is determined by measuring the time of occurrence of each cycle of the downconverted signal.

An internal digital divider performs the function of the gate generator, producing a measurement gate that always opens on the pulse's second IF cycle. The end of the gate, however, is variable and programmable via the HP 5361A's front panel or HP-IB. The counter is initially programmed to set a gate width equal to exactly three cycles of the downconverted signal, and to record the length of time this gate is open. (Note that the number of cycles (three) divided by this time gives the downconverted frequency for the beginning of the pulse.)

The gate width is then increased to contain four IF cycles, and the new time interval measured. This measurement process is continued, with the gate width systematically increased by one IF cycle for each measurement, until the end of the pulse is reached. (For larger pulse widths, the user may want to step the gate width by more than one IF cycle. The program described in this note chooses the number of IF cycles such that 100 data points can be obtained.) This process is illustrated in Figure 2.




Figure 2. Measuring the time of each IF cycle by using event gating.

Each integral number of cycles thus has a corresponding time measurement. This table of time and events (cycles of the downconverted frequency) is stored manually or in a computer, and converted to frequency by simple subtraction and division.

Frequency Determination

The frequency at any point in time can be found easily and accurately by looking at the time required for one complete IF cycle at that point. Frequency is just the reciprocal of this time.

For example, the frequency at a point 100 nsec into a pulse is found by first locating 100 nsec in the stored table and noting the corresponding number of cycles (n). One cycle is subtracted, and the time corresponding to this new number of cycles is found (t_{n-1}). The downconverted frequency is then just the number of events (one) divided by the elapsed time ($t_n - t_{n-1}$).*

Minimum Equivalent Gate Width of 10.5 nSec

Because the counter can record time for each cycle of the downconverted signal frequency, this method can provide a data point for frequency determination on every cycle. The HP 5361A downconverts the unknown signal to between 45 MHz and 95 MHz (nominally centered about 70 MHz). Thus, the measurement gate time will be between (1/95 MHz) and (1/45 MHz), or between 10.5 nsec and 22.2 nsec (14 nsec nominal).

Simple Computer Control

■ A short HP BASIC computer program can be written to automate Built-In Profiling and to perform the simple data reduction. For convenience, a sample program can be found at the end of this Note or ordered with the enclosed reply card. This program provides the gating control for the HP 5361A, collects and analyzes the data, and graphs the resulting frequency profile. In addition, optional digital filtering to smooth the data and improve resolution can be implemented.

Versatile Profiling

■ The HP 5361A is the first microwave frequency counter that offers the versatility of pulse profiling with conventional techniques (using a delaying pulse generator) as well as the new Built-In Profiling technique. For a highly accurate, simple method of profiling repetitive frequency chirps, steps, and pulses, try the HP 5361A and its new Built-In Profiling capabilities.

* See the HP 5361A Demo/Getting Started Guide (5952-7985) for details on using Scope-View to see the gating process.

Automated Built-In Profiling Sample Program

```

10 DIM Man_freq$(80),Ans1$(80),Events(101),Times(101),Hi(101),Lo(101)
20 DIM Freq(100),Freq2(100),If2(101),Freq3(100),If3(101),If1(101),SS(24)
30 COM /Profile_data/ Freq_max,Freq_min,Ans,Stop_val,Increment_value
40 ASSIGN @Hp5361a TO 814
50 No_of_events =3
60 Increment_value=1
70 Stop_val =100
80 Freq_min =MAXREAL
90 Freq_max =MINREAL
100 SS="" *** PROFILING ***
110 !
120 OUTPUT @Hp5361a;"INIT" ! INITIALIZE THE HP 5361A
130 OUTPUT @Hp5361a;"DISPLAY,"&SS ! DISPLAY "PROFILING" MESSAGE
140 OUTPUT @Hp5361a;"MMOD,PULSE" ! LOCK INTO PULSE MEAS MODE
150 OUTPUT @Hp5361a;"SAMPLE,HOLD" ! PUT INTO SAMPLE HOLD MODE
160 OUTPUT @Hp5361a;"CHIRP,ON" ! TURN ON CHIRP MODE
170 TRIGGER @Hp5361a ! TRIGGER A MEAS. THE FIRST MEAS
180 WAIT 1 ! IN 'HOLD' MODE IN NOT ALWAYS GOOD
190 TRIGGER @Hp5361a ! SO TRIGGER AGAIN
200 ENTER @Hp5361a;Ans ! GET THE AVERAGE FREQUENCY FOR SETTING
210 ! MANUAL MODE
220 OUTPUT @Hp5361a;"PAVER,1" ! SETUP TO GET THE NUMBER OF EVENTS
230 OUTPUT @Hp5361a;"EFUN,33"
240 TRIGGER @Hp5361a
250 OUTPUT @Hp5361a;"EFUN?"
260 ENTER @Hp5361a;Ans1$
270 Events(1)=VAL(Ans1$(2,12)) ! EXTRACT THE NUMBER OF EVENTS
280 IF (Events(1)<100) THEN ! IF IT IS LESS THAN 100, USE THAT NUMBER
290 Stop_val=Events(1)
300 ELSE ! OR USE 100 POINTS WITH A LARGER STEP SIZE
310 Increment_value=Events(1) DIV 100
320 END IF
330 !
340 OUTPUT @Hp5361a;"PAVER,3000" ! SETUP TO PROFILE THE PULSE
350 Man_freq$="MANUAL,"&VAL$(Ans) ! USE THE FREQUENCY FOUND ABOVE
360 OUTPUT @Hp5361a;Man_freq$
370 OUTPUT @Hp5361a;"EFUN,33" ! SETUP TO READ THE EVENTS AND TIMES
380 !
390 FOR I=1 TO Stop_val+1 ! RUN THE MAIN LOOP
400 OUTPUT @Hp5361a;"GWID,"&VAL$(No_of_events)&","EVENT"
410 OUTPUT @Hp5361a;"TRIGGER"
420 OUTPUT @Hp5361a;"EFUN?"
430 ENTER @Hp5361a;Ans1$
440 Events(I)=VAL(Ans1$(2,12))
450 Times(I)=VAL(Ans1$(14,24))
460 !
470 ! IF THE NUMBER OF EVENTS PROGRAMMED ARE GREATER THAN 47 AND DIVISIBLE
480 ! BY 16, THEN THE COUNTER CHOOSES ONE LESS EVENT FOR THE GATE WIDTH
490 ! e.g. NUMBER OF EVENTS = 64
500 ! COUNTER CHOOSES 63 EVENTS
510 !
520 IF (I>40) THEN ! INTERPOLATE FOR GATE WIDTH MENTIONED ABOVE
530 IF NOT ((I>=40) AND ((I+2) MOD 16)=1) THEN
540 IF (Times(I)=Times(I-1)) THEN GOTO 400 ! RARE, BUT COULD HAPPEN
550 Events(I-1)=(Events(I)+Events(I-2))/2
560 Times(I-1)=(Times(I)+Times(I-2))/2
570 END IF
580 END IF
590 !
600 No_of_events=No_of_events+Increment_value
610 NEXT I

```

```

620 !
630 ! RUN DATA THROUGH SMOOTHING FILTER
640 Lo(1)=1
650 Hi(1)=Stop_val
660 Kmax=1
670 Filter_f5(lf2(*),Lo(*),Hi(*),Events(*),Times(*),Kmax) ! FIVE POLE FILTER
680 Filter_f7(lf3(*),Lo(*),Hi(*),Events(*),Times(*),Kmax) ! SEVEN POLE FILTER
690 !
700 FOR l=2 TO Stop_val+1           ! CALCULATE 'l' FROM EVENTS AND TIME
710   lf1(l-1)=(Events(l)-Events(l-1))/(Times(l)-Times(l-1))*1.E+9
720 NEXT l
730 !
740 OUTPUT @Hp5361a;"EFUN,4"       ! DETERMINE THE LOCAL OSCILLATOR FREQUENCY
750 TRIGGER @Hp5361a
760 OUTPUT @Hp5361a;"EFUN?"
770 ENTER @Hp5361a;Ans1$
780 Local_osc_freq=VAL(Ans1$[3])*100000
790 !
800 OUTPUT @Hp5361a;"EFUN,68"     ! DETERMINE THE HARMONIC NUMBER AND THE
810 TRIGGER @Hp5361a             ! SIDEBAND
820 OUTPUT @Hp5361a;"EFUN?"
830 ENTER @Hp5361a;Ans1$
840 Harm_num=VAL(Ans1$[12])/100
850 N_lo=Harm_num*Local_osc_freq
860 !
870 IF Ans1$[18,18]="L" THEN ! CALCULATE THE FREQUENCY FOR THE LOWER
880   FOR l=1 TO Stop_val         ! SIDEBAND
890     Freq(l)=N_lo-lf1(l)
900     Freq2(l)=N_lo-lf2(l)
910     Freq3(l)=N_lo-lf3(l)
920     IF (Freq(l)<Freq_min) THEN Freq_min=Freq(l)
930     IF (Freq(l)>Freq_max) THEN Freq_max=Freq(l)
940   NEXT l
950 ELSE                         ! OR THE UPPER SIDEBAND
960   FOR l=1 TO Stop_val
970     Freq(l)=N_lo+lf1(l)
980     Freq2(l)=N_lo+lf2(l)
990     Freq3(l)=N_lo+lf3(l)
1000    IF (Freq(l)<Freq_min) THEN Freq_min=Freq(l)
1010    IF (Freq(l)>Freq_max) THEN Freq_max=Freq(l)
1020  NEXT l
1030 END IF
1040 !
1050 Freq_max=(INT(Freq_max/1.E+5)+1)*1.E+5
1060 Freq_min=(INT(Freq_min/1.E+5)-1)*1.E+5
1070 !
1080 OUTPUT @Hp5361a;"DISPLAY,""
1090 OUTPUT @Hp5361a;"INIT"
1100 LOCAL @Hp5361a
1110 !
1120 Graph_it(Freq(*))
1130 !
1140 PRINT "PRESS CONTINUE TO PLOT THE FILTERED DATA ( FIVE POLES )"
1150 PAUSE
1160 Graph_it(Freq2(*))          ! GRAPH THE FIVE POLE FILTERED DATA
1170 !
1180 PRINT "PRESS CONTINUE TO PLOT THE FILTERED DATA ( SEVEN POLES )"
1190 PAUSE
1200 Graph_it(Freq3(*))        ! GRAPH THE SEVEN POLE FILTERED DATA
1210 !
1220 END
1230 !
1260 SUB Graph_it(Freq(*))
1270   COM /Profile_data/ Freq_max,Freq_min,Ans,Stop_val,Increment_value
1280   IF ((Freq_max-Ans)>(Ans-Freq_min)) THEN

```

```

1290 Max_deviation=Freq_max-Ans
1300 ELSE
1310 Max_deviation=Ans-Freq_min
1320 END IF
1330 !
1340 SELECT Max_deviation/1.E+4 ! AUTO SCALE THE FREQUENCY AXIS
1350 CASE <10
1360 Division=1.0E+3
1370 Div2=1.E+6
1380 CASE <100
1390 Division=1.E+4
1400 Div2=1.E+5
1410 CASE ELSE
1420 Division=1.E+5
1430 Div2=1.E+4
1440 END SELECT
1450 !
1460 Max_deviation=INT(Max_deviation/Division)*Division
1480 GINIT ! SETUP TO PLOT THE DATA
1490 GCLEAR
1500 GRAPHICS ON
1510 PLOTTER IS CRT,"INTERNAL"
1520 !
1530 ! LABEL THE GRAPHS
1540 !
1550 LORG 4
1560 FOR I=1 TO 1.6 STEP .1
1570 MOVE 60+I,65
1580 LABEL "PULSE PROFILE"
1590 NEXT I
1600 DEG
1610 LDIR 90
1620 LORG 6
1630 CSIZE 3.5
1640 MOVE 0,40
1650 LABEL "FREQUENCY (GHz)"
1660 LORG 4
1670 LDIR 0
1680 MOVE 60,3
1690 LABEL "TIME (nSec)"
1700 !
1710 ! SETUP THE DRAWING AREA
1720 !
1730 VIEWPORT 15,115,10,60
1740 FRAME
1750 WINDOW 1,Stop_val,Freq_min,Freq_max
1760 GRID 1,Max_deviation/10,-2,Ans,10,10,0
1770 !
1780 ! DRAW THE DATA !!
1790 !
1800 MOVE 1,Freq(1)
1810 FOR I=1 TO Stop_val
1820 IF NOT ((I>=40) AND (((I+3) MOD 16)=0)) THEN PLOT I,Freq(I)
1830 NEXT I
1840 !
1850 ! NUMBER AXIS
1860 !
1870 CLIP OFF
1880 CSIZE 2.6,.6
1890 LORG 8
1900 FOR I=Ans TO Freq_max STEP Max_deviation/5
1910 MOVE 1,I
1920 LABEL USING "#,DD.DDDD";(INT(I/Division)/Div2)
1930 NEXT I
1940 FOR I=Ans TO Freq_min STEP -Max_deviation/5

```

```

1950 MOVE 1,I
1960 LABEL USING "#,DD.DDDD";(INT(I/Division)/Div2)
1970 NEXT I
1980 !
1990 LORG 6
2000 FOR I=8 TO Stop_val STEP 10
2010 MOVE I,Freq_min-1
2020 LABEL USING "#,DDDD";(I+2)*(Increment_value*14)
2030 NEXT I
2040 SUBEND
2070 !
2080 SUB Filter_f5(Freq(*),Lo(*),Hi(*),E(*),T(*),Kmax)
2090 Filter_f5:! Freq(*) computed over 5 points, less at the edges
2100 INTEGER I,L,H,K
2110 IF Kmax<1 THEN SUBEXIT
2120 FOR K=1 TO Kmax
2130 L=Lo(K)
2140 H=Hi(K)
2150 Freq(L)=(E(L+1)-E(L))/(T(L+1)-T(L))*1.E+9
2160 Freq(H)=(E(H)-E(H-1))/(T(H)-T(H-1))*1.E+9
2170 IF H-L>2 THEN
2180 Freq(L+1)=(E(L+2)-E(L))/(T(L+2)-T(L))*1.E+9
2190 Freq(H-1)=(E(H)-E(H-2))/(T(H)-T(H-2))*1.E+9
2200 IF H-L>4 THEN
2210 De1=E(L+3)-E(L)
2220 Dt1=T(L+3)-T(L)
2230 FOR I=L+2 TO H-2
2240 De2=E(I+2)-E(I-1)
2250 Dt2=T(I+2)-T(I-1)
2260 Freq(I)=(De2+De1)/(Dt2+Dt1)*1.0E+9
2270 De1=De2
2280 Dt1=Dt2
2290 NEXT I
2300 END IF
2310 END IF
2320 NEXT K
2330 SUBEND !Filter_f5
2340 !
2350 SUB Filter_f7(Freq(*),Lo(*),Hi(*),E(*),T(*),Kmax)
2360 Filter_f7:!
2370 INTEGER I,L,H,K,J
2380 IF Kmax<1 THEN SUBEXIT
2390 FOR K=1 TO Kmax
2400 L=Lo(K)
2410 H=Hi(K)
2420 Freq(L)=(E(L+1)-E(L))/(T(L+1)-T(L))*1.E+9
2430 Freq(H)=(E(H)-E(H-1))/(T(H)-T(H-1))*1.E+9
2440 IF H-L>2 THEN
2450 Freq(L+1)=(E(L+2)-E(L))/(T(L+2)-T(L))*1.E+9
2460 Freq(H-1)=(E(H)-E(H-2))/(T(H)-T(H-2))*1.E+9
2470 IF H-L>4 THEN
2480 FOR J=L TO H-4 STEP H-L-4
2490 De1=E(J+3)-E(J)
2500 Dt1=T(J+3)-T(J)
2510 De2=E(J+4)-E(J+1)
2520 Dt2=T(J+4)-T(J+1)
2530 Freq(J+2)=(De1+De2)/(Dt1+Dt2)*1.E+9
2540 NEXT J
2550 IF H-L>6 THEN
2560 De1=E(L+4)-E(L)
2570 De2=E(L+5)-E(L+1)
2580 Dt1=T(L+4)-T(L)

```



```
2590 Dt2=T(L+5)-T(L+1)
2600 FOR I=L+3 TO H-3
2610 De3=E(I+3)-E(I-1)
2620 Dt3=T(I+3)-T(I-1)
2630 Freq(I)=(De1+De2+De3)/(Dt1+Dt2+Dt3)*1.E+9
2640 De1=De2
2650 De2=De3
2660 Dt1=Dt2
2670 Dt2=Dt3
2680 NEXT I
2690 END IF
2700 END IF
2710 END IF
2720 NEXT K
2730 SUBEND !Filter_f7
```

For more information, call your local HP sales office listed in your telephone directory or an HP regional office listed below for the location of your nearest sales office.

United States:

Hewlett-Packard Company
4 Choke Cherry Road
Rockville, MD 20850
(301) 670-4300

Hewlett-Packard Company
5201 Tollview Dr.
Rolling Meadows, IL 60008
(312) 255-9800

Hewlett-Packard Company
5161 Lankershim Blvd.
No. Hollywood, CA 91601
(818) 505-5600

Hewlett-Packard Company
2015 South Park Place
Atlanta, GA 30339
(404) 955-1500

Canada:

Hewlett-Packard Ltd.
6877 Goreway Drive
Mississauga, Ontario L4V1M8
(416) 678-9430

Japan:

Yokogawa-Hewlett-Packard Ltd.
29-21, Takaido-Higashi 3-chome
Suginami-ku, Tokyo 168
(03) 331-6111

Latin America:

Latin American Region Headquarters
Monte Pelvoux Nbr 111
Lomas De Chapultepec
11000 Mexico, D.F. Mexico
(905) 596-79-33

Australia/New Zealand:

Hewlett-Packard Australia Ltd.
31-41 Joseph Street
Blackburn, Victoria 3103
Melbourne, Australia
(03) 895-2895

Far East:

Hewlett-Packard Asia Ltd.
22/F Bond Centre
West Tower
89 Queensway
Central, Hong Kong
(5) 8487777

Germany:

Hewlett-Packard GmbH
Vertriebzentrale Deutschland
Hewlett-Packard-Strasse
Postfach 1641
6380 Bad Homburg v.d.H.
Federal Republic of Germany
06172/400-0

France:

Hewlett-Packard France
Parc d'activite du Bois Briard
2, avenue du Lac
91040 Evry Cedex
01/60/77 83 83

United Kingdom:

Hewlett-Packard Ltd.
Customer Information Centre
King Street Lane
Winnersh
Wokingham
Berkshire
RG115AR
0734777828

Italy:

Hewlett-Packard Italiana S.p.A.
Via G. diVittorio, 9
20063 Cernusco Sul Naviglio (MI)
Milano
02/923691

European Multi Country Regi

Hewlett-Packard S.A.
Route du Nant d'Avril 150
1217 Meyrin 2-Geneva
Switzerland
(41) 22/83 81 11

Or write to:

United States:

Hewlett-Packard Company
P.O. Box 10301,
Palo Alto, CA 94303-0890

Europe/Middle East/Africa:

Hewlett-Packard Company
Central Mailing Department
P.O. Box 529
1180 AM Amstelveen
The Netherlands

For all other areas:

Hewlett-Packard Company
International Headquarters
3495 Deer Creek Rd.
Palo Alto, CA 94304
U.S.A.



Printed in U.S.A.
Data Subject to Change
5952-7991
August 1989